Prepare

Some parameters to be used

In[317]:=
```
SetDirectory@NotebookDirectory[]
imgSize = Large
```

Out[317]=
```
/Users/leima/GitHub/WhyMathematica/Physics/andersonLocalization
```

Out[318]=
```
Large
```

Anderson Localization Demonstration

This notebook demonstrates the Anderson Localization using MatrixPlot or ArrayPlot.

Define Parameters

Define the dimonsion of the matrix

In[319]:=
```
dim = 200;
```

Construct Matrices

Construct two matrices, one with random tridiagonal elements the other with 0.1 for second diagonal elements.

In[320]:=
```
matRandom = SparseArray[{Band[{2, 1}] → RandomReal[0.1, dim - 1],
     Band[{1, 1}] → 1., Band[{1, 2}] → RandomReal[0.1, dim - 1]}, dim];
```

In[321]:=
```
matRegular =
  SparseArray[{Band[{2, 1}] → 0.1, Band[{1, 1}] → 1., Band[{1, 2}] → 0.1}, dim];
```

Show the matrix form of these matrices

In[322]:=
```
matRandom // MatrixForm;
```

In[323]:=
```
matRegular // MatrixForm;
```

Plot the matrix themselves

In[324]:=
```
MatrixPlot[matRandom, ImageSize → imgSize]
```

Out[324]=



In[325]:=
```
MatrixPlot[matRegular, ImageSize → imgSize]
```

Out[325]=



Find Eigen Vectors

Find the eigen vectors of the matrices

In[326]:=
```
eigVRand = Transpose@Eigenvectors[matRandom] // Quiet;
% // MatrixForm;
```

In[328]:=
```
eigVReg = Transpose@Eigenvectors[matRegular] // Quiet;
% // MatrixForm;
```

Plot Eigen Vectors

Define frame labes

In[330]:=
```
frameLabel = {"# of Elements in Eigen Vectors", "# of Eigen Vectors"};
```

Plot the eigen vectors

In[331]:=
```
pltRand = ArrayPlot[eigVRand, PlotLegends → Automatic,
   ImageSize → imgSize, FrameLabel → frameLabel, FrameTicks -> Automatic]
```

Out[331]=



In[332]:=
```
pltReg = ArrayPlot[eigVReg, PlotLegends → Automatic,
  FrameTicks -> Automatic, FrameLabel → frameLabel, ImageSize → imgSize]
```

Out[332]=



Export Images

In[333]:=
```
Export["pltRand.png", pltRand]
```

Out[333]=
```
pltRand.png
```

In[334]:=
```
Export["pltReg.png", pltReg]
```

Out[334]=
```
pltReg.png
```

Acknowledgement

Thanks to Professor Cahill at University of New Mexico for explaining the idea of Anderson localization

to me.
http://theory.phys.unm.edu/cahill/